

    

        ash_circuit_breaker

        v0.1.0



    


  

    Table of contents

    
      



      	README


      	Change Log


      	LICENSE


      	AshCircuitBreaker





        	
          Modules
          


      	AshCircuitBreaker


      	AshCircuitBreaker.Change


      	AshCircuitBreaker.Info


      	AshCircuitBreaker.Preparation


      	AshCircuitBreaker.Transformer





    	Exceptions
      


      	AshCircuitBreaker.CircuitBroken



      

    




        



      

    

  

    README

[image: Ash CI]
[image: License: MIT]
[image: Hex version badge]
[image: Hexdocs badge]
AshCircuitBreaker
Welcome! This is an extension for the Ash framework
which protects your application from cascading failures by adding circuit breaker functionality to actions.
Uses the excellent fuse library to provide robust circuit breaker features that help your application gracefully handle and recover from failures.
Installation
Add ash_circuit_breaker to your list of dependencies in mix.exs:
def deps do
  [
    {:ash_circuit_breaker, "~> 0.1.0"}
  ]
end
Quick Start
	Add to your resource: Use the circuit DSL section in your Ash resource:

defmodule MyApp.Post do
  use Ash.Resource,
    domain: MyApp,
    extensions: [AshCircuitBreaker]

  circuit do
    # Protect create action - open circuit after 5 failures in 30 seconds
    action :create,
      limit: 5,
      per: :timer.seconds(30),
      reset_after: :timer.minutes(5)

    # Protect update action with different thresholds
    action :update,
      limit: 10,
      per: :timer.minutes(1),
      reset_after: :timer.minutes(2)

    # Only break circuit for specific error types
    action :delete,
      limit: 3,
      per: :timer.seconds(15),
      reset_after: :timer.minutes(5),
      should_break?: fn error ->
        # Only break on timeout or database errors, ignore validation errors
        match?(%{class: :timeout}, error) or match?(%{class: :database}, error)
      end
  end

  # ... rest of your resource definition
end
	That's it! Your actions are now protected by circuit breakers. When failures exceed the threshold, subsequent calls will be blocked with an AshCircuitBreaker.CircuitBroken error until the circuit resets.

Basic Usage
Simple Circuit Breaker
circuit do
  # Open circuit after 5 failures in 30 seconds, reset after 5 minutes
  action :create,
    limit: 5,
    per: :timer.seconds(30),
    reset_after: :timer.minutes(5)
end
Multiple Actions
circuit do
  action :create, limit: 5, per: :timer.seconds(30), reset_after: :timer.minutes(5)
  action :update, limit: 10, per: :timer.minutes(1), reset_after: :timer.minutes(2)
  action :delete, limit: 3, per: :timer.seconds(15), reset_after: :timer.minutes(10)
end
Error Filtering
circuit do
  # Only break circuit on infrastructure errors, not validation errors
  action :create,
    limit: 5,
    per: :timer.seconds(30),
    reset_after: :timer.minutes(5),
    should_break?: fn error ->
      # Don't break on user input validation errors
      not match?(%Ash.Error.Invalid{}, error)
    end

  # Never break circuit for analytics actions
  action :track_event,
    limit: 100,
    per: :timer.minutes(1),
    reset_after: :timer.seconds(30),
    should_break?: fn _error -> false end
end
Custom Circuit Names
circuit do
  # Use a custom static name
  action :create,
    limit: 5,
    per: :timer.seconds(30),
    reset_after: :timer.minutes(5),
    name: :my_custom_circuit

  # Use a function to generate dynamic names
  action :update,
    limit: 10,
    per: :timer.minutes(1),
    reset_after: :timer.minutes(2),
    name: fn changeset ->
      :"update_circuit_#{changeset.data.id}"
    end
end
Advanced Usage
Manual Integration
For more control, you can add circuit breaker protection directly to specific actions:
defmodule MyApp.Post do
  use Ash.Resource, domain: MyApp

  actions do
    create :create do
      change {AshCircuitBreaker.Change,
        limit: 5,
        per: :timer.seconds(30),
        reset_after: :timer.minutes(5)}
    end

    update :update do
      change {AshCircuitBreaker.Change,
        limit: 10,
        per: :timer.minutes(1),
        reset_after: :timer.minutes(2),
        should_break?: fn error ->
          # Only break on system errors, not validation errors
          not match?(%Ash.Error.Invalid{}, error)
        end}
    end
  end
end
Custom Name Functions
The name function determines how circuit breakers are identified and shared:
# Per-user circuit breakers
name: fn changeset, context ->
  :"user_#{context.actor.id}_create"
end

# Per-tenant circuit breakers
name: fn changeset ->
  :"tenant_#{changeset.data.tenant_id}_action"
end

# Use the built-in name function (default)
name: &AshCircuitBreaker.name_for_breaker/1
Error Filtering with should_break?
By default, any error will trigger the circuit breaker. Use should_break? to only break on specific error types:
circuit do
  # Only break on timeout errors
  action :create,
    limit: 5,
    per: :timer.seconds(30),
    reset_after: :timer.minutes(5),
    should_break?: fn error ->
      match?(%{class: :timeout}, error)
    end

  # Break on any error (default behavior when should_break? is not specified)
  action :delete,
    limit: 3,
    per: :timer.seconds(15),
    reset_after: :timer.minutes(5),
    should_break?: fn _error -> true end
end
Circuit Breaker States
A circuit breaker can be in one of two states:
	OK (normal operation): Requests flow through normally. Failures are counted until the limit is reached.
	Blown (failing fast): All requests are immediately rejected with CircuitBroken error until the reset period expires.

Error Handling
When a circuit is open, an AshCircuitBreaker.CircuitBroken exception is raised:
case MyApp.create_post(attrs) do
  {:ok, post} ->
    # Success
    {:ok, post}

  {:error, %AshCircuitBreaker.CircuitBroken{} = error} ->
    # Circuit breaker is open
    {:error, "Service temporarily unavailable, please try again later"}

  {:error, other_error} ->
    # Handle other errors (these may trigger the circuit breaker)
    {:error, other_error}
end
In web applications, the exception includes Plug.Exception behaviour for automatic HTTP 503 responses.
Configuration
Parameters
	limit: Maximum number of failures allowed before opening the circuit
	per: Time window (in milliseconds) for counting failures
	reset_after: Time (in milliseconds) before attempting to close an open circuit
	name: Identifier for the circuit breaker (atom or function)
	should_break?: Function that takes an error and returns true if the circuit should break, false otherwise (optional, defaults to breaking on any error)

Example Configurations
# High-traffic endpoint with tight failure tolerance
action :api_call, limit: 3, per: :timer.seconds(10), reset_after: :timer.seconds(30)

# Background job with more lenient settings
action :process_data, limit: 20, per: :timer.minutes(5), reset_after: :timer.minutes(10)

# Critical operation with long recovery time
action :payment, limit: 1, per: :timer.seconds(5), reset_after: :timer.minutes(30)
Error Filtering with should_break?
The should_break? option gives you fine-grained control over which errors should trigger the circuit breaker. This is particularly useful when you want to:
	Distinguish between user errors and system errors: Don't break the circuit for validation failures caused by bad user input
	Ignore non-critical errors: Keep the circuit closed for operations where failures are acceptable
	Target specific error types: Only break on infrastructure issues like timeouts or database connection failures

Common Error Filtering Patterns
# Only break on timeout errors
should_break?: fn error ->
  match?(%{class: :timeout}, error)
end

# Only break on database/infrastructure errors
should_break?: fn error ->
  match?(%{class: :database}, error) or
  match?(%{class: :timeout}, error) or
  match?(%{class: :network}, error)
end

# Ignore validation errors (user input problems)
should_break?: fn error ->
  not match?(%Ash.Error.Invalid{}, error)
end

# Only break on specific error codes
should_break?: fn error ->
  case error do
    %{code: code} when code in ["TIMEOUT", "CONNECTION_REFUSED", "SERVICE_UNAVAILABLE"] -> true
    _ -> false
  end
end

# Custom logic based on error message
should_break?: fn error ->
  message = Exception.message(error)
  String.contains?(message, "database") or String.contains?(message, "timeout")
end
Conditional Circuit Breaking
# Break only during business hours
should_break?: fn error ->
  hour = DateTime.utc_now().hour
  # More strict during business hours (9 AM - 5 PM UTC)
  if hour >= 9 and hour <= 17 do
    true  # Break on any error during business hours
  else
    # Only break on severe errors outside business hours
    match?(%{class: :timeout}, error)
  end
end
Monitoring
Circuit breaker state can be monitored using fuse's built-in functions:
# Check circuit state
:fuse.ask(:my_circuit_name, :sync)
# Returns: :ok | :blown | {:error, :not_found}

# Reset a blown circuit manually
:fuse.reset(:my_circuit_name)

# Melt (blow) a circuit manually
:fuse.melt(:my_circuit_name)
Testing
In test environments, you may want to disable circuit breakers or use test-friendly configurations:
# Use very high limits in tests
circuit do
  action :create, limit: 999, per: :timer.hours(1), reset_after: :timer.seconds(1)
end

# Or conditionally apply circuit breakers
if Mix.env() != :test do
  circuit do
    action :create, limit: 5, per: :timer.seconds(30), reset_after: :timer.minutes(5)
  end
end
Limitations
	Circuit breakers are not supported for read/query actions - they only work with create, update, and delete actions
	Circuit breaker state is not persisted across application restarts
	Fuse circuits are local and don't share state across nodes in a distributed system

Reference
	AshCircuitBreaker DSL
	Fuse Documentation
	Circuit Breaker Pattern



  

    Change Log

All notable changes to this project will be documented in this file.
See Conventional Commits for commit guidelines.
v0.1.0 (2025-08-02)
Features:
	add should_break? option to help ignore certain errors

v0.0.1 (2025-08-02)
Initial release!


  

    LICENSE

MIT License
Copyright (c) 2025 Christian Alexander
Permission is hereby granted, free of charge, to any person obtaining a copy
of this software and associated documentation files (the "Software"), to deal
in the Software without restriction, including without limitation the rights
to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
copies of the Software, and to permit persons to whom the Software is
furnished to do so, subject to the following conditions:
The above copyright notice and this permission notice shall be included in all
copies or substantial portions of the Software.
THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN THE
SOFTWARE.


  

    AshCircuitBreaker

An extension for Ash.Resource which adds the ability to wrap actions in circuit breakers to allow for graceful handling of and recovery from failures.
circuit
Configure a circuit breaker for actions.
Note that this extension does not support circuit breaking for read/query actions.
Fuse
This library uses the fuse package to provide
circuit breaker features. See fuse's documentation for more information.
Names
Fuse uses an atom "name" value as a reference to a specific circuit breaker. By default, the name is derived from the action name and resource name. You can provide a custom name using the name option.
Nested DSLs
	action

circuit.action
action action
Configure a circuit breaker for a single action.
It does this by adding a global change or preparation to the resource with the provided configuration.
Arguments
	Name	Type	Default	Docs
	action	atom		The name of the action to wrap in a circuit breaker

Options
	Name	Type	Default	Docs
	limit	pos_integer		The maximum number of failures allowed before the circuit opens
	per	pos_integer		The time period (in milliseconds) for which failures are counted
	reset_after	pos_integer		The time period (in milliseconds) after which the circuit will attempt to close again
	name	atom | (any -> any) | (any, any -> any)	&AshCircuitBreaker.name_for_breaker/1	The name to use for the circuit breaker. This can be an atom or a function that takes a query/changeset and optional context object to generate an atom key.
	should_break?	nil | (any -> any)		A function that takes the error and returns true if the circuit should break. If not provided, the circuit will break on any error.

Introspection
Target: AshCircuitBreaker

An extension for Ash.Resource which adds the ability to wrap actions in circuit breakers to allow for graceful handling of and recovery from failures.

          @type error_matcher() :: (any() -> boolean())

          @type namefun() ::
  (Ash.Changeset.t() | Ash.ActionInput.t() -> atom())
  | (Ash.Changeset.t() | Ash.ActionInput.t(), map() -> atom())

          @type t() :: %AshCircuitBreaker{
  __identifier__: any(),
  action: atom(),
  limit: pos_integer(),
  name: atom() | namefun(),
  per: pos_integer(),
  reset_after: pos_integer(),
  should_break?: error_matcher() | nil
}

Generates a name for the circuit breaker based on the action input or changeset.

A resource change which implements a circuit breaker.

Auto-generated introspection functions.

          @spec circuit(dsl_or_extended :: module() | map()) :: [struct()]

A resource preparation which implements a circuit breaker.

A Spark DSL transformer for circuit breaking.

An exception which is raised or returned when an action invocation is blocked by a circuit breaker.

        Create an Elixir.AshCircuitBreaker.CircuitBroken without raising it.

          @spec exception(opts :: Keyword.t()) :: %AshCircuitBreaker.CircuitBroken{
  __exception__: true,
  action: term(),
  bread_crumbs: term(),
  class: term(),
  limit: term(),
  path: term(),
  per: term(),
  reset_after: term(),
  splode: term(),
  stacktrace: term(),
  vars: term()
}


      


Create an Elixir.AshCircuitBreaker.CircuitBroken without raising it.
Keys
	:action
	:limit
	:per
	:reset_after

